



Spesifikasi:

Ukuran: 14x21 cm

Tebal: 270 hlm

Harga: Rp 44.800

Terbit pertama: Juni 2005

Sinopsis singkat:

Buku ini secara khusus hadir untuk melengkapi buku sejenis yang telah ada dengan memfokuskan penggunaan berbagai fungsi dan fasilitas yang dimiliki program aplikasi Microsoft Excel. Mengedepankan pembahasan 83 studi kasus terapan berbasis komputer untuk membantu pembaca dalam mengelola dana yang dimilikinya. Materi bahasan ditulis sistematis dengan bahasa yang lugas, jelas, mudah, singkat, dan sederhana. Ditujukan kepada semua kalangan yang berminat mempelajari perencanaan keuangan secara benar dan terarah. Tanpa perlu belajar secara khusus dan berlatar belakang pendidikan tertentu, Anda dapat langsung mengoperasikannya.

Buku ini disertai pula dengan CD berisi file contoh studi kasus yang dapat dimodifikasi sesuai dengan kebutuhan. Setiap bahasan studi kasus dilengkapi media latihan dengan arahan penyelesaian dan ilustrasi layar untuk memudahkan Anda mempelajari materi tanpa harus selalu berhadapan dengan komputer sehingga proses belajar menjadi efektif dan efisien.

BAB 2

ANGSURAN PINJAMAN

Bunga dan Angsuran Pinjaman

Salah satu masalah yang sering dihadapi oleh peminjam baik pribadi maupun perusahaan dalam meminjam dana kepada pihak lain adalah menghitung angsuran pinjaman. Hal ini penting berkaitan dengan kemampuan pribadi maupun perusahaan dalam rangka melunasi kewajiban dalam periode waktu tertentu. Bahasan bab ini akan mengupas sistem perhitungan angsuran, rincian pembayaran tahunan dan pengambilan keputusan jika terdapat tawaran pinjaman dengan bunga yang lebih rendah. Dasar teori yang digunakan dalam bahasan bab ini mengacu pada materi yang banyak diungkap dalam buku manual manajemen keuangan. Fokus bahasan ditujukan untuk memberikan pemahaman tentang berbagai teori yang mendukung perhitungan angsuran pinjaman. Setelah memahami materi bahasan, diharapkan pembaca mendapatkan gambaran hasil perhitungan angsuran pinjaman sebagai dasar untuk perencanaan keuangan baik dari pihak peminjam maupun pemberi pinjaman.

Angsuran pinjaman yang harus dibayar oleh peminjam dipengaruhi oleh pokok pinjaman, jangka waktu pinjam, dan tingkat suku bunga yang berlaku. Besar bunga pinjaman yang harus dibayar oleh peminjam dapat dihitung berdasarkan sistem yang dipakai yaitu bunga tetap (*flat*), menurun (*sliding*), dan efektif. Salah satu fungsi finansial yang disediakan program aplikasi Microsoft Excel dapat digunakan untuk menghitung bunga dan angsuran pinjaman dengan sistem atau metode efektif. Berbeda dengan perhitungan dengan metode efektif, penggunaan metode bunga tetap dan bunga menurun mengharuskan Anda menyusun rumus atau formula. Perhatikan penjelasan tentang hal tersebut dalam subbahasan berikut ini.

Bunga Tetap

Perhitungan bunga kredit sistem Flat atau model *constant payment mortgage* menghasilkan angsuran pinjaman yang tetap dari periode ke periode (bulan). Hal ini disebabkan bunga dihitung tetap dari pokok pinjaman awal, secara matematis dapat dihitung dengan rumus sebagai berikut:

$$= (\text{SALDO PINJAMAN AWAL} \times \text{BUNGA PER TAHUN}) / 12$$

Bunga Menurun

Perhitungan bunga kredit sistem Sliding atau model *adjusted rate mortgage* akan menghasilkan bunga yang semakin menurun, dengan demikian angsuran pinjaman dari periode ke periode juga akan menurun. Perhitungan bunga didasarkan pada saldo pinjaman yang semakin mengecil, secara matematis dapat dihitung dengan rumus sebagai berikut:

$$= (\text{SALDO PINJAMAN} \times \text{BUNGA PER TAHUN}) / 12$$


Bunga Efektif

Besar angsuran pinjaman dengan perhitungan bunga efektif adalah tetap (seperti sistem Flat), tetapi cicilan pokok pinjaman menaik dan bunga per bulan menurun (seperti sistem Sliding). Perhitungan angsuran pinjaman dengan metode ini berbeda dengan bunga tetap dan menurun. Excel telah menyediakan fungsi untuk menghitung Bunga Pinjaman, Cicilan Pokok Pinjaman, Pembayaran Angsuran, Bunga Pinjaman Kumulatif, dan Cicilan Pokok Pinjaman Kumulatif.

Tabel Angsuran Pinjaman

Bahasan materi bab ini akan mengupas pembuatan tabel angsuran pinjaman dengan bunga tetap, menurun, efektif, rincian pembayaran dalam periode tahunan dan pemilihan alternatif pinjaman yang dipengaruhi oleh tingkat suku bunga. Materi bahasan tersimpan dalam buku kerja atau file BAB02, dengan nama lembar kerja (sheet) KASUS. Setiap materi yang dibahas, disertai dengan media untuk berlatih yang tersimpan dalam sheet LATIH.

Untuk memudahkan dalam mempelajari bahasan tertentu, disediakan sheet MENU. Anda dapat mengaktifkan sheet tertentu dengan klik pada daftar pilihan yang tersedia, selanjutnya sheet terpilih aktif. Bagian atas tampilan sheet terpilih telah disediakan tombol navigasi, untuk kembali ke sheet MENU, mengaktifkan sheet sebelah kiri atau sebelah kanan dari posisi sheet aktif.



Gambar 2.1. Tampilan tombol pilihan dalam sheet MENU

Studi Kasus

Materi bahasan studi kasus tersimpan dalam buku kerja atau file BAB02, terdiri dari 12 lembar kerja atau sheet. Terdiri dari enam sheet pembahasan kasus dengan nama KASUS dan enam sheet yang disediakan untuk berlatih Anda dengan nama LATIH. Materi bahasan disertai dengan tabel angsuran dibuat dengan fasilitas format kondisional (*conditional formatting*) sehingga baris dalam tabel mengikuti jangka waktu pinjam. Jangka waktu pinjam dalam bahasan materi bab ini bervariasi antara 24 bulan atau 2 tahun sampai dengan 60 bulan (5 tahun). Jika Anda akan menambah data jangka waktu pinjam, diharapkan dapat mempelajari prosedur penggunaan fasilitas format kondisional yang dibahas dalam bab ini. Asumsi yang digunakan dalam bahasan bab ini, program aplikasi Excel dan buku kerja BAB02 telah aktif. Selanjutnya Anda dapat belajar sesuai dengan topik yang dikehendaki melalui studi kasus berikut ini.

Studi Kasus 1 – Perhitungan Angsuran Pinjaman dengan Bunga Tetap

Sebuah perusahaan merencanakan untuk meminjamkan uang kepada seorang karyawan sebesar Rp 30 juta. Tingkat suku bunga pinjaman ditetapkan sebesar 18% per tahun dengan jangka waktu pinjam 12 bulan. Perhitungan angsuran pinjaman disepakati dengan bunga tetap, dan dibayarkan setiap akhir bulan. Berdasarkan data tersebut, dapat dibuat tabel yang memuat informasi antara lain bulan pembayaran, pokok pinjaman, cicilan pokok pinjaman, bunga, angsuran per bulan, dan saldo pokok pinjaman.

Prosedur penyelesaian studi kasusnya sebagai berikut:

1. Pilih dan klik tab sheet KASUS1 (Anda dapat mengikuti bahasan melalui sheet LATIH1) atau dengan klik salah satu tombol dalam sheet MENU.
2. Kolom Bulan Ke akan terisi angka berurutan mulai dari 1 sampai dengan angka dalam isian jangka waktu pinjam, dibuat dengan fungsi IF berikut ini:
 - Sel B15 =IF(G7=0;"";1)
 - Sel B16 =IF(B15="";"";IF(G\$7>=B15+1;B15+1;""))

Artinya, pada alamat sel B15 diisi angka 1 jika isian jangka waktu pinjam (G7) tidak sama dengan nol. Jika isian alamat sel G7 adalah nol, isian sel B15 dikosongkan. Alamat sel B16 diisi dengan menjumlahkan isian pada alamat B15 ditambah 1, jika nilai sel G7 (jangka waktu pinjam) lebih besar sama dengan isian pada alamat sel B15 ditambah 1.
3. Kolom Bulan (pembayaran angsuran) diisi dengan ketetapan angsuran dibayar setiap akhir bulan, mulai dari bulan pinjam. Pengisian kolom ini menggunakan kombinasi fungsi IF dan EOMONTH berikut ini:
 - Sel C15 =IF(B15="";"";EOMONTH(G9;0))
 - Sel C16 =IF(B16="";"";EOMONTH(G\$9;B15))

Artinya, pada alamat sel diisi dengan akhir bulan pinjaman, dalam kasus ini pinjaman tanggal 3 Mei 2005 (G9) sehingga pembayaran pertama pada akhir Mei 2005. Isian sel C6, pembayaran kedua dilakukan berdasarkan tanggal pinjaman (G9) dengan interval waktu satu bulan (B15) setelah tanggal pinjam.

The screenshot shows an Excel spreadsheet titled "ANGSURAN PINJAMAN PERHITUNGAN BUNGA TETAP". It includes a summary table and a detailed monthly payment table.

BUNGA PINJAMAN per Tahun	18%
Jangka Waktu Pinjam	12 bulan
Pokok Pinjaman	30.000.000
Tanggal Pinjam	3-Mei-2005
Angsuran per Bulan	2.950.000

BULAN KE	BULAN	POKOK PINJAMAN	CICILAN POKOK PINJAMAN	BUNGA	ANGSURAN PER BULAN	SALDO POKOK PINJAMAN
1	Mei 2005	30.000.000	2.500.000	450.000	2.950.000	27.500.000
2	Jun 2005	27.500.000	2.500.000	450.000	2.950.000	25.000.000
3	Jul 2005	25.000.000	2.500.000	450.000	2.950.000	22.500.000
4	Agust 2005	22.500.000	2.500.000	450.000	2.950.000	20.000.000
5	Sep 2005	20.000.000	2.500.000	450.000	2.950.000	17.500.000
6	Okt 2005	17.500.000	2.500.000	450.000	2.950.000	15.000.000
7	Nop 2005	15.000.000	2.500.000	450.000	2.950.000	12.500.000
8	Des 2005	12.500.000	2.500.000	450.000	2.950.000	10.000.000
9	Jan 2006	10.000.000	2.500.000	450.000	2.950.000	7.500.000
10	Feb 2006	7.500.000	2.500.000	450.000	2.950.000	5.000.000
11	Mar 2006	5.000.000	2.500.000	450.000	2.950.000	2.500.000
12	Apr 2006	2.500.000	2.500.000	450.000	2.950.000	0

Gambar 2.2. Perhitungan Angsuran Pinjaman dengan Bunga Tetap

Asumsi yang digunakan dalam kasus ini, angsuran pinjaman akan dipotong dari gaji karyawan yang dibayar setiap akhir bulan sehingga tanggal akhir bulan tidak ditampilkan. Jika Anda menghendaki tanggal akhir bulan ditampilkan, silakan format lengkap (dd mmm yyy) pada kolom yang bersangkutan. Atau, jika tanggal pembayaran angsuran pinjaman pertama (di akhir bulan) ditetapkan satu bulan setelah pinjam. Isian fungsi pada alamat sel C15 menjadi

$$=IF(B15="", "", EOMONTH(G9;B15))$$

Fungsi EOMONTH (*end of month*) digunakan untuk menentukan tanggal akhir bulan dari sebuah tanggal awal dengan interval waktu tertentu (sekian bulan lagi). Bentuk penulisan fungsi adalah sebagai berikut:

$$=EOMONTH(start_date;months)$$

start_date diisi dengan tanggal awal (dapat diwakili oleh alamat sel yang berisi tanggal tersebut) sebagai dasar perhitungan.

months diisi angka yang menunjukkan interval waktu.

Misal, pada alamat sel B15 terdapat isian tanggal 10 Agustus 2005 dan Anda dapat mengetahui akhir bulan ke-5 dari tanggal tersebut jatuh pada tanggal dan bulan tertentu. Penulisan fungsi adalah =EOMONTH(B15;5) dengan hasil 31 Januari 2006.

4. Pokok Pinjaman diisi dengan fungsi IF berikut ini:

- Sel D15 =IF(B15="";"";G8)
- Sel D16 =IF(B16="";"";H15)

Artinya, pokok pinjaman pada bulan ke-1 (D15) mengacu pada pokok pinjaman yang terdapat pada alamat sel G8. Pokok pinjaman bulan berikutnya (mulai dari D16) mengacu pada saldo pokok pinjaman (mulai dari H15).

5. Cicilan Pokok Pinjaman (mulai dari E15) dilakukan dengan membagi nilai Pokok Pinjaman (G8) dengan Jangka Waktu Pinjam (G7) dengan fungsi berikut ini:

=IF(B15="";"";G\$8/G\$7)

Data pada alamat sel G8 dan G7 selanjutnya akan disalin ke bawah, sehingga di depan nomor baris alamat sel ditambahkan tanda \$.

6. Bunga Pinjaman (mulai dari F15) dihitung berdasarkan perkalian antara Saldo Pinjaman Awal (G8) dengan Bunga pinjaman per tahun (G6). Selanjutnya hasil dibagi 12, perhatikan penulisan fungsi berikut ini:

=IF(B15="";"";(G\$8*G\$6)/12)

7. Angsuran Pinjaman (mulai dari G15) merupakan hasil penjumlahan dari Cicilan Pokok Pinjaman (E15) dengan Bunga (F15), dengan fungsi berikut ini:

=IF(B15="";"";E15+F15) atau =IF(B15="";"";SUM(E15:F15))

Nilai angsuran pinjaman relatif sama dari periode ke periode, sehingga pengisian data Angsuran per Bulan pada alamat sel G11 diisi dengan fungsi =IF(G7=0;"";G15).

8. Saldo Pokok Pinjaman (mulai dari H15) dihitung berdasarkan nilai Pokok Pinjaman (G8) dikurangi dengan jumlah Cicilan Pokok Pinjaman (mulai dari E15) dengan penulisan fungsi berikut ini:


=IF(B15="";"";G\$8-SUM(E\$15:E15))

9. Salin fungsi yang terdapat pada range B16:D16 dan tempatkan hasil (dalam kasus ini) pada range B17:D38. Selanjutnya salin fungsi yang terdapat pada range E15:H15 dan tempatkan hasil pada range E16:H38.

Penempatan bingkai secara otomatis yang terdapat pada range B15:H38 dibuat menggunakan fasilitas format kondisional dengan prosedur sebagai berikut:

1. Sorot atau blok range B15:H38.
2. Kemudian pilih dan klik menu **Format > Conditional Formatting**, jendela Conditional Formatting ditampilkan.
 - Condition 1 klik drop-down dan pilih Cell Value Is.
 - Operator logika pilih dan klik operator logika between, selanjutnya tentukan nilai 0 (nol) dan $=G\$8+(\$G\$6*\$G\$8)$.
 - Klik tombol Format dan tentukan format bingkai yang dikehendaki.

Gambar berikut menunjukkan tampilan jendela Conditional Formatting yang telah diisi.



Gambar 2.3. Tampilan jendela/kotak dialog Conditional Formatting

Anda dapat melihat tampilan jendela tersebut dengan menempatkan penunjuk sel pada suatu sel dalam area range B15:H38 dan klik menu **Format > Conditional Formatting**.

3. Klik tombol OK.

Studi Kasus 2 – Perhitungan Angsuran Pinjaman dengan Bunga Menurun

Studi kasus kedua pada prinsipnya hampir sama dengan studi kasus pertama yaitu peminjaman dana sebesar Rp 30 juta dengan tingkat suku bunga 18% per tahun dengan jangka waktu 12 bulan. Perbedaan terletak pada sistem

bunga yang dijadikan dasar untuk perhitungan angsuran yaitu bunga menurun. Hasil perhitungan angsuran akan menunjukkan jumlah yang semakin menurun dari periode ke periode pembayaran (bulanan) seperti dijelaskan berikut ini.

Prosedur penyelesaian studi kasusnya sebagai berikut:

1. Pilih dan klik tab sheet KASUS2 (Anda dapat mengikuti bahasan melalui sheet LATIH2) atau klik salah satu tombol yang terdapat dalam sheet MENU.
2. Bunga Pinjaman (mulai dari F15) dihitung berdasarkan perkalian antara Pokok Pinjaman pada bulan berjalan (D15) dengan Bunga pinjaman per tahun (G6). Selanjutnya hasil dibagi 12, perhatikan penulisan fungsi berikut ini:

$$=IF(B15="";"";(D15*G$6)/12)$$

The screenshot shows an Excel spreadsheet with the following data:

ANGSURAN PINJAMAN		Bunga Pinjaman per Tahun	18%
		Jangka Waktu Pinjam	12 bulan
		Pokok Pinjaman	30.000.000
		Tanggal Pinjam	3-Mei-2005
		Angsuran Bulan ke-5	2.800.000

TABEL ANGSURAN PINJAMAN						
BULAN KE	BULAN	POKOK PINJAMAN	CICILAN POKOK PINJAMAN	BUNGA	ANGSURAN PER BULAN	POKOK PINJAMAN
1	Mei 2005	30.000.000	2.500.000	450.000	2.950.000	27.500.000
2	Jun 2005	27.500.000	2.500.000	412.500	2.912.500	25.000.000
3	Jul 2005	25.000.000	2.500.000	375.000	2.875.000	22.500.000
4	Agust 2005	22.500.000	2.500.000	337.500	2.837.500	20.000.000
5	Sep 2005	20.000.000	2.500.000	300.000	2.800.000	17.500.000
6	Okt 2005	17.500.000	2.500.000	262.500	2.762.500	15.000.000
7	Nop 2005	15.000.000	2.500.000	225.000	2.725.000	12.500.000
8	Des 2005	12.500.000	2.500.000	187.500	2.687.500	10.000.000
9	Jan 2006	10.000.000	2.500.000	150.000	2.650.000	7.500.000
10	Feb 2006	7.500.000	2.500.000	112.500	2.612.500	5.000.000
11	Mar 2006	5.000.000	2.500.000	75.000	2.575.000	2.500.000
12	Apr 2006	2.500.000	2.500.000	37.500	2.537.500	0


Gambar 2.4. Perhitungan Angsuran Pinjaman dengan Bunga Menurun

Angsuran pinjaman per bulan dari periode ke periode semakin menurun, Anda dapat melihat data tersebut dengan menempatkan penunjuk sel pada E11. Lihat dan perhatikan gambar berikut ini:

BULAN KE	BULAN	POKOK PINJAMAN	ANGSURAN PER BULAN	SALDO POKOK PINJAMAN
1	Mei 2005	30.000.000	2.800.000	27.500.000
2	Jun 2005	27.500.000	2.800.000	25.000.000
3	Jul 2005	25.000.000	2.800.000	22.500.000
4	Agust 2005	22.500.000	2.800.000	20.000.000
5	Sep 2005	20.000.000	2.800.000	17.500.000

Gambar 2.5. Pemilihan Periode Angsuran Pinjaman

Pilihan untuk mengisi Jangka Waktu Pinjam dan Angsuran Bulan ke-.. dibuat menggunakan fasilitas validasi data. Anda dapat melihat proses pembuatan pilihan dengan pilih dan klik menu **Data > Validation**, lihat Gambar 2.6 berikut:



Gambar 2.6. Tampilan jendela/kotak dialog Data Validation

Daftar isian telah dibuat dengan nama range BULAN dan BULAN1 yang terdapat pada range I23:J47 dalam sheet KASUS4. Gambar tersebut menunjukkan tampilan kotak dialog atau jendela Data Validation setelah Anda menempatkan penunjuk sel di E11 dan klik menu **Data > Validation**.

Angka yang menunjukkan nilai Angsuran bulan ke-.. pada alamat sel G11, diisi menggunakan kombinasi fungsi IF dan VLOOKUP berikut ini:

=IF(G7=0;"";VLOOKUP(E11;ANGSUR;6))

Studi Kasus 3 – Perhitungan Angsuran Pinjaman dengan Bunga Efektif

Data studi kasus ketiga pada dasarnya sama dengan data pada studi kasus sebelumnya yaitu tentang pinjaman sebesar Rp 30 juta dengan tingkat suku bunga 18% per tahun dan jangka waktu pengembalian 12 bulan. Perhitungan angsuran pinjaman dalam studi kasus ketiga menggunakan sistem bunga efektif, seperti dijelaskan melalui bahasan berikut ini dan Gambar 2.7.

Prosedur penyelesaian studi kasusnya sebagai berikut:

1. Pilih dan klik tab sheet KASUS3 (Anda dapat mengikuti bahasan melalui sheet LATIH3) atau dengan klik salah satu tombol dalam sheet MENU.
2. Cicilan Pokok Pinjaman (mulai dari sel E15) dihitung dengan fungsi PPMT, untuk otomatisasi tabel, selanjutnya dikombinasikan dengan fungsi IF berikut ini:

$$=IF(B15="";";PPMT(G$6/12;B15;G$7;-G$8;1))$$

3. Bunga (mulai dari sel F15) dihitung dengan fungsi IPMT berikut ini:

$$=IF(B15="";";IPMT(G$6/12;B15;G$7;-G$8;1))$$

The screenshot shows an Excel spreadsheet titled "ANGSURAN PINJAMAN PERHITUNGAN BUNGA EFEKTIF". It contains a summary table and a detailed monthly payment table.

Bunga Pinjaman per Tahun		18%
Jangka Waktu Pinjam		12 bulan
Pokok Pinjaman		30.000.000
Tanggal Pinjam		3-Mei-2005
Angsuran per Bulan		2.750.400

BULAN KE	BULAN	POKOK PINJAMAN	CICILAN POKOK PINJAMAN	BUNGA	ANGSURAN PER BULAN	SALDO POKOK PINJAMAN
1	Mei 2005	30.000.000	2.300.400	450.000	2.750.400	27.699.600
2	Jun 2005	27.699.600	2.334.906	415.494	2.750.400	25.364.695
3	Jul 2005	25.364.695	2.369.929	380.470	2.750.400	22.994.765
4	Agust 2005	22.994.765	2.405.478	344.921	2.750.400	20.589.287
5	Sep 2005	20.589.287	2.441.560	308.839	2.750.400	18.147.727
6	Okt 2005	18.147.727	2.478.184	272.216	2.750.400	15.669.543
7	Nop 2005	15.669.543	2.515.357	235.043	2.750.400	13.154.186
8	Des 2005	13.154.186	2.553.087	197.313	2.750.400	10.601.099
9	Jan 2006	10.601.099	2.591.383	159.016	2.750.400	8.009.716
10	Feb 2006	8.009.716	2.630.254	120.146	2.750.400	5.379.462
11	Mar 2006	5.379.462	2.669.708	80.692	2.750.400	2.709.754
12	Apr 2006	2.709.754	2.709.753	40.646	2.750.400	1

Gambar 2.7. Perhitungan Angsuran Pinjaman dengan Bunga Efektif

4. Angsuran per Bulan (mulai dari sel G15) dihitung dengan fungsi PMT berikut ini:

=IF(B15="", "", PMT(G\$6/12;G\$7;-G\$8))

Studi Kasus 4 – Perhitungan Bunga dan Cicilan Pinjaman Kumulatif dengan Bunga Efektif

Amir - seorang karyawan, pada awal bulan Oktober 2005 akan mendapatkan pinjaman dari perusahaan sebesar Rp 10 juta dengan jangka waktu pelunasan selama 6 bulan. Tingkat suku bunga pinjaman yang diberlakukan sebesar 12% per tahun dengan sistem bunga efektif. Untuk keperluan perencanaan keuangan di akhir tahun, kedua belah pihak (Amir dan perusahaan) menghendaki informasi berapa besar bunga dan angsuran pinjaman kumulatif berdasarkan periode yang telah ditetapkan.

Solusi perhitungan bunga kumulatif dan pinjaman kumulatif menggunakan fungsi yang telah disediakan oleh Excel. Fungsi CUMIPMT (*cumulative interest payment*) digunakan untuk menghitung bunga kumulatif, sedangkan fungsi CUMPRINC (*cumulative principle*) untuk menghitung pokok pinjaman kumulatif.

Prosedur penyelesaian studi kasusnya sebagai berikut:

1. Pilih dan klik tab sheet KASUS4 (Anda dapat mengikuti bahasan melalui sheet LATIH4) atau klik salah satu tombol dalam sheet MENU.
2. Tetapkan awal periode dan akhir periode untuk perhitungan kumulatif (bunga dan cicilan pokok pinjaman) pada alamat sel D16 dan E16.
3. Bunga Pinjaman Kumulatif (F16) dihitung dengan fungsi sebagai berikut:

=-CUMIPMT(G6/12;G7;G8;D16;E16;0)

4. Cicilan Pokok Kumulatif (G16) dihitung dengan fungsi sebagai berikut:

=-CUMPRINC(G6/12;G7;G8;D16;E16;0)

Perhatikan hasilnya melalui gambar berikut ini:

ANGSURAN PINJAMAN						
PERHITUNGAN BUNGA EFEKTIF						
Bunga Pinjaman per Tahun	12%					
Jangka Waktu Pinjam	6 bulan					
Pokok Pinjaman	10.000.000					
Tanggal Pinjam	2-Oct-2005					
Angsuran per Bulan	1.725.484					
Kumulatif Bunga dan Cicilan Pokok Pinjaman						
Periode Bulan	Kumulatif					
Awal	Akhir	Bunga	Cicilan			
1	3	251.073	4.925.378			
TABEL ANGSURAN PINJAMAN						
BULAN KE	BULAN	POKOK PINJAMAN	CICILAN POKOK PINJAMAN	BUNGA	ANGSURAN PER BULAN	SALDO POKOK PINJAMAN
1	Okt 2005	10.000.000	1.625.484	100.000	1.725.484	8.374.516
2	Nop 2005	8.374.516	1.641.738	83.745	1.725.484	6.732.778
3	Des 2005	6.732.778	1.658.156	67.328	1.725.484	5.074.622
4	Jan 2006	5.074.622	1.674.737	50.746	1.725.484	3.399.885
5	Feb 2006	3.399.885	1.691.485	33.999	1.725.484	1.708.400
6	Mar 2006	1.708.400	1.708.399	17.084	1.725.484	1

Gambar 2.8. Perhitungan Bunga dan Cicilan Pokok Pinjaman Kumulatif dengan Bunga Efektif

Studi Kasus 5 – Angsuran Pinjaman dan Rincian Pembayaran

Studi kasus kelima pada dasarnya adalah pengembangan dari studi kasus pertama sampai dengan ketiga yang digabung dalam satu lembar kerja. Yaitu sebuah lembar kerja yang dapat menampung dan digunakan untuk perhitungan angsuran pinjaman dengan sistem bunga tetap, menurun serta efektif. Pemilihan sistem bunga dilakukan dengan klik tombol pilihan (*option button*) disertai dengan rincian pembayaran bunga, cicilan pokok pinjaman, dan angsuran pada masing-masing tahun pembayaran. Jumlah pinjaman yang dilakukan relatif besar sehingga jangka waktu pinjam sampai dengan 5 tahun dan terdapat periode tenggang waktu pembayaran (*grace period*).

Misal, Alan seorang karyawan mendapatkan pinjaman dari perusahaan tempat dia bekerja sebesar Rp 300 juta, jangka waktu pengembalian 5 tahun dengan tenggang waktu pembayaran 2 bulan. Tingkat suku bunga pinjaman disepakati sebesar 12% per tahun dan berlaku sampai jangka waktu pinjaman. Pinjaman tersebut akan direalisasi pada tanggal 5 Januari 2005 dan sesuai perhitungan akan lunas dalam bulan Februari 2010. Untuk perencanaan keuangan kedua belah pihak, sistem perhitungan bunga pinjaman nantinya dipilih dari ketiga sistem yaitu tetap (*flat*), menurun atau efektif.

Selain itu, kedua belah pihak juga memerlukan rincian pembayaran atau penerimaan mulai dari tahun 2005 sampai dengan 2010.

Prosedur penyelesaian studi kasusnya sebagai berikut:

1. Pilih dan klik tab sheet KASUS5 (Anda dapat mengikuti bahasan melalui sheet LATIH5) atau dengan cara klik salah satu tombol dalam sheet MENU. Sebagian data ditampilkan melalui Gambar 2.9. berikut ini:

BUNGA PINJAMAN per Tahun	12,00%					
Jangka Waktu Pinjam	60 bulan					
Pokok Pinjaman	300.000.000					
Tanggal Pinjam	5-Jan-2005					
Tenggang Waktu Pembayaran	2 bulan					
Angsuran per Bulan 8.000.000						
TABEL ANGSURAN PINJAMAN						
BULAN KE	BULAN	POKOK PINJAMAN	CICILAN POKOK PINJAMAN	BUNGA	ANGSURAN PER BULAH	SALDO POKOK PINJAMAN
1	Mar 2005	300.000.000	5.000.000	3.000.000	8.000.000	295.000.000
2	Apr 2005	295.000.000	5.000.000	3.000.000	8.000.000	290.000.000
3	Mei 2005	290.000.000	5.000.000	3.000.000	8.000.000	285.000.000
4	Jun 2005	285.000.000	5.000.000	3.000.000	8.000.000	280.000.000
5	Jul 2005	280.000.000	5.000.000	3.000.000	8.000.000	275.000.000
6	Agust 2005	275.000.000	5.000.000	3.000.000	8.000.000	270.000.000
7	Sep 2005	270.000.000	5.000.000	3.000.000	8.000.000	265.000.000
8	Oktr 2005	265.000.000	5.000.000	3.000.000	8.000.000	260.000.000
9	Nop 2005	260.000.000	5.000.000	3.000.000	8.000.000	255.000.000
10	Des 2005	255.000.000	5.000.000	3.000.000	8.000.000	250.000.000
11	Jan 2006	250.000.000	5.000.000	3.000.000	8.000.000	245.000.000
12	Feb 2006	245.000.000	5.000.000	3.000.000	8.000.000	240.000.000
13	Mar 2006	240.000.000	5.000.000	3.000.000	8.000.000	235.000.000
14	Apr 2006	235.000.000	5.000.000	3.000.000	8.000.000	230.000.000
15	Mei 2006	230.000.000	5.000.000	3.000.000	8.000.000	225.000.000

Gambar 2.9. Perhitungan Angsuran Pinjaman dan Rincian Pembayaran

2. Klik salah satu pilihan sistem perhitungan bunga pinjaman (tetap, efektif atau menurun).

Tombol pilihan dibuat menggunakan toolbar Forms pilihan Option Button (aktifkan melalui menu View > Toolbars > Forms, selanjutnya pilih dan klik Option Button. Pilihan tombol akan menghasilkan angka 1, 2 dan 3 yang disebut dengan pengaturan Format Control dan pada kasus ini ditempatkan pada alamat sel B11. Angka hasil pemilihan tersebut untuk perhitungan cicilan pokok pinjaman dan bunga pinjaman seperti dibahas berikut ini.

3. Bulan (C15) yang merupakan pembayaran angsuran dimulai dari bulan ke-1 (B15) dipengaruhi oleh Tanggal Pinjam (E8) dan Tenggang Waktu Pembayaran (E9). Dengan asumsi pembayaran dilakukan pada akhir bulan yang telah ditentukan, bulan pembayaran ditetapkan dengan fungsi EOMONTH berikut ini:

$$=IF(B15="";"";EOMONTH(E$8;C$11+B15))$$

- Sel E8 merupakan isian tanggal pinjam.
 - Sel C11 alamat sel sembarang untuk penempatan angka sebagai pembantu interval waktu yang berasal dari tenggang waktu pembayaran (E9) dikurangi 1. Angka 1 sebagai pengurang, sesuai dengan tenggang waktu yang pada akhirnya akan ditambahkan dengan angka yang menunjukkan bulan pembayaran ke- mulai dari sel B15, dengan rumus =E9-1.
 - Sel B15 urutan bulan pembayaran yang merupakan interval waktu perhitungan pembayaran. Lihat bahasan studi kasus 1.
4. Cicilan Pokok Pinjaman (E15) khusus untuk pilihan bunga Efektif (menghasilkan angka 2) dihitung dengan fungsi PMT, pilihan lain dihitung berdasarkan Pokok Pinjaman dibagi Jangka Waktu Pinjam dengan fungsi sebagai berikut:

$$=IF(B15="";"";IF(B$11=2;PPMT(E$5/12;B15;E$6;-E$7;1);E$7/E$6))$$

5. Bunga Pinjaman (F15) dihitung sesuai dengan pilihan sistem bunga (penjelasan dapat dibaca pada bahasan sebelumnya) dengan fungsi sebagai berikut:

$$=IF(B15="";"";IF(B$11=1;(E$7*E$5/12);IF(B$11=2;IPMT(E$5/12;B15;E$6;-E$7);(D15*E$5/12))))$$

6. Rincian Pembayaran berisi rincian pembayaran (atau penerimaan bagi pihak yang meminjamkan dana) pada masing-masing tahun berupa cicilan pokok pinjaman, bunga, dan angsuran. Bagian ini juga dilengkapi dengan pembayaran secara keseluruhan seperti ditunjukkan melalui Gambar 2.10 berikut ini:

RINCIAN PEMBAYARAN			
TOTAL PEMBAYARAN			
	CICILAN	BUNGA	JUMLAH
	POKOK PINJAMAN		ANGSURAN
	300.000.000	91.500.000	391.500.000
RINCIAN PEMBAYARAN			
TAHUN	CICILAN	BUNGA	JUMLAH
	POKOK PINJAMAN		ANGSURAN
2005	60.000.000	27.750.000	77.750.000
2006	60.000.000	26.700.000	86.700.000
2007	60.000.000	19.500.000	79.500.000
2008	60.000.000	12.300.000	72.300.000
2009	60.000.000	5.100.000	65.100.000
2010	10.000.000	150.000	10.150.000

Gambar 2.10. Rincian Pembayaran Cicilan Pokok Pinjaman, Bunga dan Jumlah Angsuran

Anda perhatikan Gambar 2.10, isian tahun pembayaran akan tampil secara otomatis dimulai dari tahun awal pembayaran (dalam kasus ini 2005) sampai dengan akhir periode pembayaran (2010). Angka tersebut ditampilkan dari tahun pembayaran yang terdapat pada isian bulan pembayaran (mulai dari alamat sel C15). Tampilan isian tahun pembayaran pada sheet KASUS5 sengaja disamarkan, Anda dapat melihat tampilan tersebut pada sheet LATIH5.

Fungsi untuk menyusun tabel Rincian Pembayaran sebagai berikut:

- Sel A15 dan seterusnya digunakan untuk menampilkan data tahun pembayaran dengan fungsi,

$$=IF(C15="";"";YEAR(C15))$$

- Sel A14 atau alamat sel lain (bebas) untuk menetapkan tahun terakhir pembayaran atau tahun terbesar dengan fungsi MAX berikut $=MAX(A16:A74)$.

Hasil perhitungan kedua fungsi di atas, digunakan untuk pembuatan isian kolom Tahun seperti dibahas berikut ini.

- Sel J24 isian awal tahun pembayaran – dapat dilakukan dengan menyalin data yang terdapat pada alamat sel A15 dengan fungsi $=IF(A15="";"";A15)$.
- Sel J25 isian tahun pembayaran ke-2 sampai dengan tahun pembayaran terakhir (dilakukan dengan menyalin

fungsi, dalam kasus ini sampai alamat sel J29), menggunakan fungsi sebagai berikut:

=IF(J24="";"";IF(J24+1<=A\$14;J24+1;""))

- Sel K24 dan seterusnya – untuk pengisian Cicilan Pokok Pinjaman dengan fungsi SUMIF berikut ini:

=IF(J24="";"";SUMIF(\$A\$15:\$A\$134;J24;\$E\$15:\$E\$134))

- Sel L24 dan seterusnya – untuk pengisian Bunga dengan fungsi SUMIF berikut ini:

=IF(J24="";"";SUMIF(\$A\$15:\$A\$134;J24;\$F\$15:\$F\$134))

- Sel M24 dan seterusnya – untuk pengisian Jumlah Angsuran dengan fungsi =IF(J24="";"";SUM(K24:L24))

Selanjutnya salin fungsi yang terdapat pada range K24:M24 dan tempatkan hasil (dalam kasus ini) ke dalam range K25:M29.

Total Pembayaran dilakukan dengan menjumlah data pada kolom yang sama dalam Rincian Pembayaran dengan fungsi SUM berikut ini:

- Sel K19 Cicilan Pokok Pinjaman dengan fungsi sebagai berikut:

=SUM(K24:K29)

- Sel L19 Bunga diisi dengan fungsi =SUM(L24:L29).


- Sel M19 Jumlah Angsuran dengan fungsi =SUM(M24:M29).

Studi Kasus 6 – Pengalihan Kredit AntarBank

PT XYZ (atau pribadi) saat ini memiliki saldo pinjaman dari sebuah bank sebesar Rp 2,75 milyar dengan waktu yang tersisa untuk pelunasan 3 tahun. Tingkat suku bunga pinjaman (*flat*) per tahun 15% dan jika melunasi pinjaman sebelum jatuh tempo dikenakan denda (*penalty*) per tahun sebesar 2%. Saat ini PT XYZ mendapatkan tawaran dari sebuah bank lain dengan tingkat bunga pinjaman sebesar 12,5% per tahun. Dengan asumsi hanya mempertimbangkan faktor finansial apakah tawaran pengalihan kredit antar bank tersebut diterima atau sebaiknya ditolak? Perhatikan jawaban melalui bahasan berikut ini.

Prosedur penyelesaian studi kasusnya sebagai berikut:

1. Pilih dan klik tab sheet KASUS6 (Anda dapat mengikuti bahasan melalui sheet LATIH6) atau dengan klik salah satu tombol dalam sheet MENU.
2. Hitung denda dan bunga jika beralih kredit (dari Bank A ke Bank B) dengan rumus dan fungsi berikut ini:
 - Sel D13 Denda – dihitung dari hasil perkalian antara jumlah kredit, sisa waktu pelunasan dan denda dengan rumus $=D8*G7*G6$.
 - Sel D14 Bunga – dihitung berdasarkan data pinjaman bank baru (dalam kasus ini disebut Bank B) ditetapkan berdasarkan hasil perkalian antara jumlah kredit, bunga per tahun dan waktu pelunasan dengan rumus $=G6*G7*D8$.
 - Sel D15 Total Pembayaran merupakan hasil penjumlahan antara Denda dengan Bunga dengan rumus $=D13+D14$ atau dengan fungsi $=SUM(D13:D14)$.



Gambar 2.11. Kredit Bank

3. Hitung bunga yang harus dibayar jika perusahaan tetap bertahan dengan kredit dari bank lama (Bank A).
 - Sel D19 Bunga – dihitung berdasarkan hasil perkalian antara jumlah kredit, bunga per tahun, dan waktu pelunasan dengan rumus $=D6*D7*D8$.

4. Kesimpulan (B22) dibuat dengan cara membandingkan hasil perhitungan seandainya tetap bertahan dengan kredit bank lama (Bank A) dengan hasil perhitungan jika kredit dialihkan ke bank baru (Bank B). Dasar pengambilan keputusan adalah biaya (bunga atau bunga dan denda) yang lebih kecil dengan fungsi sebagai berikut:

=IF(D19>D15;" Sebaiknya perusahaan beralih kredit pinjaman ke "&F4;"
Sebaiknya perusahaan tetap bertahan dengan kredit dari "&B4)